Constant Stress Tethers

G. Flanagan
Alna Space Program

Initial: April 20, 2011, Last Edit: July 19, 2011

Introduction

This notebook looks at the cable masses need for three scenarios. The first is a vertically hanging orbital cable with gravity gradient effects. The second reproduces the mass ratio deviation of Moravec for a spinning cable without gravity effects. The third combines a gravity gradient and the centrifugal effects. The goal in each case is to provide a skyhook that allows for payload pickup at a velocity less than the low earth orbit velocity. The goal is to pickup a payload at a velocity (relative to earth surface) that is substantially lower than the velocity required for low-earth-orbit. Thus, the lifter that meets the cable will require a need a much lower propel lent mass fraction. A 50% reduction in pickup velocity would have a substantial system cost benefit.

These systems transfer energy and momentum stored in the cable system to the payload. Because the cable is much more massive than the payload, the cable continues in a slightly lower energy orbit. The cable must be restored to its original energy before the next payload using some form of high-efficiency propulsion.

In these notes, the terms “cable” and “tether” are being used synonymously. We make the assumption that the minimum mass cable will have a constant material stress along it’s length. Thus, the cable cross-section varies with position. The equations related to the material distribution from the cable system center-of-gravity (CG) to the payload pickup point. The cable to payload mass ratio is the performance metric of interest. However, we will only account for the mass from the CG to the pickup. The total mass would include the mass above the CG, but there are a number of design choices for this remaining mass that this study does not address. The other half of the cable may also have a constant-stress shape. Alternatively, the cable may be tied to a larger mass, including the mass for power and propulsion to restore the cable to it’s original orbit after delivering a payload.

These equations are not original, but it is useful to re-derive previous results in *Mathematica*, and we will have consistent design functions available for future work. For example, see:

Moravec, Journal of the Astronautical Sciences, v25 #4, pp. 307-322, Oct-Dec 1977

Hans Moravec "Orbital Bridges," (1986)

Kirk F. Sorensen, "Conceptual Design and Analysis of an MXER Tether Boost Station", AIAA 2001-3915

Thomas J. Bogar et al, "Hypersonic Airplane Space Tether Orbital Launch (HASTOL) System: Interim Study Results", AIAA-99-4802

Simple vertical cable

Consider a vertical (relative to earth surface) cable stabilized by gravity gradient effect. Concept is to dangle a tether from high orbit, Rcg, and capture payload near earth surface (Rleo). Because the cable orbital velocity (Vcg) is lower than the orbital velocity at Rleo, the lifter does not need the full ΔV required for orbit. The payload would then use some type of climbing system to raise it’s elevation to Rcg or beyond.

For simplicity, we’ll ignore the earth’s rotational velocity. Also, a real cable would need to be in an elliptical orbit so that the slow-down at payload pickup results in an apogee decrease, and not a burn-up.



Derivation

Net acceleration at r due to gravity and centrifugal effect of orbital velocity. The origin for r is the center of earth.

In[1]:=
```
a[r_] = μ / r^2 - Vcg^2 / r
```

Out[1]= $-\frac{Vcg^2}{r}+\frac{\mu}{r^2}$

Differential equation for area, where T is the tensile capability of the material, and ρ is the material density.

In[2]:=
```
TetherArea = Area /. First[DSolve[T Area'[r] == ρ Area[r] a[r], Area, r]] // Simplify
```

Out[2]= $\text{Function}\left[\{r\},\ e^{-\frac{\mu\rho}{r T}-\frac{Vcg^2 \rho \text{Log}[r]}{T}} C[1]\right]$

The area at the capture point is just enough to support the tipMass (payload plus capture mechanism), with the acceleration at the near earth location Re

In[3]:=
```
c = First@Solve[TetherArea[Rleo] == tipMass * a[Rleo] / T, C[1]]
```

Out[3]= $\left\{C[1]\to -\frac{e^{\frac{\mu\rho}{Rleo\,T}}\, Rleo^{-2+\frac{Vcg^2\rho}{T}}\, tipMass\,(Rleo\, Vcg^2-\mu)}{T}\right\}$

Integrate to get the total tether mass up to Rcg. Also, apply a series of assumptions on the range of the physical quantities.

In[4]:=
```
mass =
  Integrate[ρ TetherArea[r] /. c, {r, 0, Rcg},
    Assumptions → {Rleo > 0, μ > 0, Rcg > 0, Rcg > Re, ρ > 0, T > 0, Vcg > 0}] // Simplify
```

Out[4]= $\frac{1}{Rleo^2\,T} e^{\frac{\mu\rho}{Rleo\,T}}\, Rcg \left(\frac{Rleo}{Rcg}\right)^{\frac{Vcg^2\rho}{T}} tipMass\,(-Rleo\,Vcg^2+\mu)\,\rho\,\text{ExpIntegralE}\left[2-\frac{Vcg^2\rho}{T},\ \frac{\mu\rho}{Rcg\,T}\right]$

Define α as the ratio of the orbital velocity of the cable cg divided by the velocity of a circular orbit at low earth orbit (Rleo). α is roughly the fraction of orbital ΔV that must be supplied by the lifter rocket. Vc is the material characteristic velocity.

In[5]:=
```
subList = Flatten@{Rcg → μ/Vcg^2, First@Solve[α == Vcg / Sqrt[μ / Rleo], Vcg], ρ / T → Vc^-2}
```

Out[5]=
$$\left\{\text{Rcg} \to \frac{\mu}{\text{Vcg}^2},\ \text{Vcg} \to \alpha \sqrt{\frac{\mu}{\text{Rleo}}},\ \frac{\rho}{\text{T}} \to \frac{1}{\text{Vc}^2}\right\}$$

In[6]:=
```
Simplify[mass //. subList]
```

Out[6]=
$$-\frac{1}{\text{Rleo Vc}^2\,\alpha^2}\, e^{\frac{\mu}{\text{Rleo Vc}^2}}\ \text{tipMass}\ \left(\alpha^2\right)^{\frac{\alpha^2 \mu}{\text{Rleo Vc}^2}} \left(-1+\alpha^2\right) \mu\ \text{ExpIntegralE}\left[2-\frac{\alpha^2 \mu}{\text{Rleo Vc}^2},\ \frac{\alpha^2 \mu}{\text{Rleo Vc}^2}\right]$$

In[7]:=
```
massRatioVerticalCable = Simplify[mass / tipMass //. subList]
```

Out[7]=
$$-\frac{1}{\text{Rleo Vc}^2\,\alpha^2}\, e^{\frac{\mu}{\text{Rleo Vc}^2}} \left(\alpha^2\right)^{\frac{\alpha^2 \mu}{\text{Rleo Vc}^2}} \left(-1+\alpha^2\right) \mu\ \text{ExpIntegralE}\left[2-\frac{\alpha^2 \mu}{\text{Rleo Vc}^2},\ \frac{\alpha^2 \mu}{\text{Rleo Vc}^2}\right]$$

In[8]:=
```
areaVerticalCable = Simplify[TetherArea[r] / tipMass /. c] //. subList
```

Out[8]=
$$\frac{e^{\frac{(\text{r}-\text{Rleo})\,\mu}{\text{r Rleo Vc}^2}}\ \text{r}^{-\frac{\alpha^2 \mu}{\text{Rleo Vc}^2}}\ \text{Rleo}^{-2+\frac{\alpha^2 \mu}{\text{Rleo Vc}^2}} \left(\mu - \alpha^2 \mu\right)}{\text{T}}$$

Examples

Gravitational constant and earth radius will be global symbols.

In[9]:=
```
unitStrip[x_] := Select[x, NumberQ]
```

In[10]:=
```
Needs["PhysicalConstants`"]
```

In[11]:=
```
earthConstants = {
  μ -> unitStrip[GravitationalConstant * EarthMass],
  Rleo → (EarthRadius // unitStrip)}
```

Out[11]=
$$\left\{\mu \to 3.98735 \times 10^{14},\ \text{Rleo} \to 6\,378\,140\right\}$$

For Spectra (see http://www51.honeywell.com/sm/afc/common/documents/PP_AFC_Honeywell_spectra_fiber_2000_Product_information_sheet.pdf). Assume a factor of safety of 2.0

In[12]:=
```
characteristicVelocity = Sqrt[T / (ρ factorSafety)];
```

In[13]:=
```
V[Spectra] = characteristicVelocity /. {ρ -> 970, T → 3. × 10^9, factorSafety → 2}
```

Out[13]=
```
1243.54
```

In[14]:=
```
massRatioVerticalCable /. earthConstants //. Vc -> V[Spectra]
```

Out[14]=
$$-\frac{1}{\alpha^2} 1.45833 \times 10^{19} \left(\alpha^2\right)^{40.4269\,\alpha^2} \left(-1+\alpha^2\right) \text{ExpIntegralE}\left[2-40.4269\,\alpha^2,\ 40.4269\,\alpha^2\right]$$

In[15]:=
```
SetOptions[Plot, GridLines → Automatic];
```

Plot the cable mass ratio for a range of velocity ratios (α). If we define “practical” as a mass ratio of less than 400, then it appears that a non-rotating cable cannot be used at a velocity ratio below about 0.77.

```
In[16]:= Plot[massRatioVerticalCable /. earthConstants //. {Vc -> V[Spectra]},
  {α, .7, .9}, GridLines → Automatic, PlotRange → {All, {0, 1000}},
  AxesLabel → {"Vel. Ratio", "Mass Ratio"}]
```

Out[16]=



Plot the shape profile for a 1000 kg tip mass.

```
In[17]:= rearth = EarthRadius // unitStrip;
```

Plot of cable cross-sectional area for a 1000 kg tip mass. Assume a velocity ratio of 0.75

```
In[18]:= Plot[1000 * areaVerticalCable /. earthConstants /. {Vc -> V[Spectra], T → 3. × 10^9} /.
    α → 0.75 /. r → (s + rearth / 1000) * 1000, {s, 0, rearth (1 / 0.75^2 - 1) / 1000},
  GridLines → Automatic, AxesLabel → {"x (km)", "Area (m^2)"}]
```

Out[18]=



Or, if you prefer, a Manipulate widget to show the same information.

In[19]:=
```
Manipulate[
 TableForm[{alpha, MaterialVelocity,
   Re[massRatioVerticalCable /. earthConstants /. {α → alpha, Vc → MaterialVelocity}]},
  TableHeadings → {{"α", "Matl. Characteristic Vel. m/sec", "Mass Ratio"}, None}],
 {{alpha, .75}, .05, 1, .05}, {{MaterialVelocity, 1200}, 1000, 10 000},
 SaveDefinitions → True, FrameLabel → "Mass Ratio for Vertical Cable"]
```

Out[19]=

alpha

MaterialVelocity

α	0.75
Matl. Characteristic Vel. m/sec	1200
Mass Ratio	986.36

Mass Ratio for Vertical Cable

Re-derive Moravec equations (rotation, no gravity)

This time, we will consider a cable that is spinning about it’s own CG. The velocity of the pickup point is a combination of the cables orbital velocity at CG along with the rotational velocity. This has been referred to as a rotovator or Moravec wheel. To match early derivations, we will first consider on the rotational accelerations and ignore the gravity gradient.

This time, the origin for r is the center of rotation for the cable.

In[20]:=
```
a[r_] = v^2 r / rtip^2;
```

Differential equation for area, where T is the tensile capability of the material, and ρ is the material density.

In[21]:=
```
A = Area /. First@DSolve[Area'[r] == -ρ / T Area[r] a[r], Area, r]
```

Out[21]= $\text{Function}\left[\{r\},\ e^{-\frac{r^2 v^2 \rho}{2\, rtip^2\, T}}\, C[1]\right]$

In[22]:=
```
c = First@Solve[A[rtip] * T == tipMass * a[rtip], C[1]]
```

Out[22]= $\left\{C[1] \to \frac{e^{\frac{v^2 \rho}{2 T}}\ tipMass\ v^2}{rtip\ T}\right\}$

In[23]:=
```
mass = Integrate[A[r] ρ, {r, 0, rtip}] /. c // Simplify
```

Out[23]= $\frac{e^{\frac{v^2 \rho}{2 T}} \sqrt{\frac{\pi}{2}}\ tipMass\ v \sqrt{\rho}\ \text{Erf}\left[\frac{v \sqrt{\rho}}{\sqrt{2} \sqrt{T}}\right]}{\sqrt{T}}$

VR is the ratio of the tip velocity to the material characteristic velocity (and, following Moravec, putting in a factor of 2 because that’s how the cluster appears in the equation above).

In[24]:=
```
tsub = First@Solve[VR == v * Sqrt[ρ / (2 T)], T]
```

Out[24]= $\left\{T \to \frac{v^2 \rho}{2\, VR^2}\right\}$

In[25]:= `massRatioMoravec = Simplify[mass / tipMass /. tsub, Assumptions → {ρ > 0, v > 0, VR > 0}]`

Out[25]= $e^{VR^2} \sqrt{\pi}\ VR\ \mathrm{Erf}[VR]$

yielding Moravec's solution.

Next, recast into orbital velocity fraction α, and using a definition of the characteristic material velocity that does not include the factor of 2 used by Moravec (to match with the form I'll use in the next section)

In[26]:= `massRatioMoravec /. VR → Vleo * α / Vc / Sqrt[2]`

Out[26]= $$\frac{e^{\frac{Vleo^2 \alpha^2}{2\,Vc^2}} \sqrt{\frac{\pi}{2}}\ Vleo\ \alpha\ \mathrm{Erf}\left[\frac{Vleo\ \alpha}{\sqrt{2}\ Vc}\right]}{Vc}$$

In[27]:= `areaMoravec = A[r] / tipMass /. c /. tsub // Simplify`

Out[27]= $$\frac{2\, e^{\left(1-\frac{r^2}{rtip^2}\right) VR^2}\ VR^2}{rtip\ \rho}$$

Plot the total mass versus the velocity ration VR

In[28]:= `Plot[massRatioMoravec, {VR, 0, 2.5}, AxesLabel → {"Velocity Ratio", "Mass Ratio"}]`

Out[28]=



Plot of the area profile, with the total length set to unity. x=0 is at the center of rotation.

In[29]:= `Plot[areaMoravec /. {rtip -> 1, VR → 2, ρ → 1}, {r, 0, 1}, AxesLabel → {"x", "Area"}]`

Out[29]=



Combined gravity gradient and centrifugal

A more accurate assessment of the rotovator mass will need to combine the gravity gradient and centrifugal acceleration. The constant stress profile will be determined at the point in the rotation where the cable is vertical relative to the earth and the accelerations directly combine.

There is no attempt at finding orbital resonances, i.e. the touchdown location for the cable will change relative to the earth's surface. The early Moravec papers attempted to use integer orbital values so that the touchdown stayed fixed. This doesn't seem practical for an oblate earth when orbital procession is considered. The lifter vehicle will need to have some operational flexibility to get to the touchdown points.



Net acceleration at r due to gravity and centrifugal effect of orbital velocity. The origin for r is the center of earth. s is the length coordinate measured from the center of rotation.

In[30]:= `a[s_] = μ / r^2 - Vcg^2 / r + Vtip^2 s / L^2 //. {r → Rcg - s, Vcg → Sqrt[μ / Rcg]} // Simplify`

Out[30]= $\frac{s\,Vtip^2}{L^2} + \frac{s\,\mu}{Rcg\,(Rcg - s)^2}$

Differential equation for area, where T is the tensile capability of the material, and ρ is the material density.

In[31]:=
```
TetherArea = Area /. First[DSolve[T Area'[s] == -ρ Area[s] a[s], Area, s]] // Simplify
```

Out[31]=
$$\text{Function}\left[\{s\},\ e^{-\frac{\rho\left(\frac{1}{2}\,\text{Rcg}\,s^2\,\text{Vtip}^2-\frac{L^2\,\text{Rcg}\,\mu}{-\text{Rcg}+s}+L^2\,\mu\,\text{Log}[-\text{Rcg}+s]\right)}{L^2\,\text{Rcg}\,T}}\ C[1]\right]$$

Solve for the required area at the tip. The length of the cable is L.

In[32]:=
```
c = Simplify@First@Solve[TetherArea[L] == tipMass * a[L] / T, C[1]]
```

Out[32]=
$$\left\{C[1]\to\left(e^{\frac{\rho\left(\text{Vtip}^2-\frac{2\mu}{L-\text{Rcg}}+\frac{2\mu\,\text{Log}[L-\text{Rcg}]}{\text{Rcg}}\right)}{2T}}\ \text{tipMass}\left(-2\,L\,\text{Rcg}^2\,\text{Vtip}^2+\text{Rcg}^3\,\text{Vtip}^2+L^2\left(\text{Rcg}\,\text{Vtip}^2+\mu\right)\right)\right)\Big/\left(L\,(L-\text{Rcg})^2\,\text{Rcg}\,T\right)\right\}$$

Take a look at the resulting area function so that we can see what manipulations can be performed to get a meaningful result.

In[33]:=
```
TetherArea[s] /. c // Simplify
```

Out[33]=
$$\left(e^{\frac{\rho\left(\text{Vtip}^2-\frac{s^2\,\text{Vtip}^2}{L^2}-\frac{2\mu}{L-\text{Rcg}}+\frac{2\mu}{-\text{Rcg}+s}+\frac{2\mu\,\text{Log}[L-\text{Rcg}]}{\text{Rcg}}-\frac{2\mu\,\text{Log}[-\text{Rcg}+s]}{\text{Rcg}}\right)}{2T}}\ \text{tipMass}\left(-2\,L\,\text{Rcg}^2\,\text{Vtip}^2+\text{Rcg}^3\,\text{Vtip}^2+L^2\left(\text{Rcg}\,\text{Vtip}^2+\mu\right)\right)\right)\Big/\left(L\,(L-\text{Rcg})^2\,\text{Rcg}\,T\right)$$

This time, I want to end up with α and L as the only design choices. Make a substitution list to get rid of other factors.

In[34]:=
```
subList =
  Flatten@{Rcg → Rleo + L, First@Solve[α Vleo == -Vtip + Vcg, Vtip], ρ / T → Vc^-2,
    Sqrt[μ / Rleo] → Vleo}
```

Out[34]=
$$\left\{\text{Rcg}\to L+\text{Rleo},\ \text{Vtip}\to\text{Vcg}-\text{Vleo}\,\alpha,\ \frac{\rho}{T}\to\frac{1}{\text{Vc}^2},\ \sqrt{\frac{\mu}{\text{Rleo}}}\to\text{Vleo}\right\}$$

In[35]:=
```
areaCombinedCable = TetherArea[s] /. c //. subList // Simplify
```

Out[35]=
$$\frac{1}{L\,(L+\text{Rleo})\,T}\,e^{\frac{(\text{Vcg}-\text{Vleo}\,\alpha)^2-\frac{s^2(\text{Vcg}-\text{Vleo}\,\alpha)^2}{L^2}+\frac{2\mu}{\text{Rleo}}-\frac{2\mu}{L+\text{Rleo}-s}}{2\,\text{Vc}^2}}\,(-\text{Rleo})^{-2+\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}$$
$$(-L-\text{Rleo}+s)^{-\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}\ \text{tipMass}\left(L\,\text{Rleo}^2\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+\text{Rleo}^3\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+L^2\,\mu\right)$$

In[36]:=
```
f = ρ areaCombinedCable / tipMass //. {ρ/T → 1/Vc^2} // Simplify
```

Out[36]=
$$\frac{1}{L\,(L+\text{Rleo})\,\text{Vc}^2}\,e^{\frac{(\text{Vcg}-\text{Vleo}\,\alpha)^2-\frac{s^2(\text{Vcg}-\text{Vleo}\,\alpha)^2}{L^2}+\frac{2\mu}{\text{Rleo}}-\frac{2\mu}{L+\text{Rleo}-s}}{2\,\text{Vc}^2}}\,(-\text{Rleo})^{-2+\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}$$
$$(-L-\text{Rleo}+s)^{-\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}\left(L\,\text{Rleo}^2\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+\text{Rleo}^3\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+L^2\,\mu\right)$$

Symbolic integration fails (note: cell has been disabled because the evaluation takes several minutes and fails to return a useful result)

```
Integrate[f, {s, 0, L}]
```

$$\int_0^L \frac{1}{L\,(L+\text{Rleo})\,\text{Vc}^2}\, e^{\frac{(\text{Vcg}-\text{Vleo}\,\alpha)^2-\frac{s^2(\text{Vcg}-\text{Vleo}\,\alpha)^2}{L^2}+\frac{2\mu}{\text{Rleo}}-\frac{2\mu}{L+\text{Rleo}-s}}{2\,\text{Vc}^2}}\,(-\text{Rleo})^{-2+\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}\,(-L-\text{Rleo}+s)^{-\frac{\mu}{(L+\text{Rleo})\,\text{Vc}^2}}\left(L\,\text{Rleo}^2\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+\text{Rleo}^3\,(\text{Vcg}-\text{Vleo}\,\alpha)^2+L^2\mu\right)ds$$

To make a function, easiest to copy/paste the above result into the function below.

In[37]:=
```
massCombinedCable[α_, L_, Vc_] := Module[{Vcg, Vleo},
  Vcg = Sqrt[μ/(L + Rleo)];
  Vleo = Sqrt[μ / Rleo];

  f = 1/(L (L + Rleo) Vc^2) E^(((Vcg - Vleo α)^2 - s^2 (Vcg - Vleo α)^2/L^2 + 2 μ/Rleo - 2 μ/(L + Rleo - s))/(2 Vc^2)) (-Rleo)^(-2 + μ/((L + Rleo) Vc^2)) (-L - Rleo + s)^(-μ/((L + Rleo) Vc^2))
    (L Rleo^2 (Vcg - Vleo α)^2 + Rleo^3 (Vcg - Vleo α)^2 + L^2 μ) /. earthConstants;

  NIntegrate[Re[f], {s, 0, L}]]
```

Plot the cable mass ratio as a function of the total cable length. The ratio starts out declining (because of the lower cg velocity), but eventually increases again because of the gravity gradient affect. The minimum for the particular material constants is for a length of approximately 1500 km.

In[38]:=
```
Plot[massCombinedCable[.5, L * 1000, V[Spectra]], {L, 10, 2 * 10^3},
 AxesLabel → {"Length (km)", "Mass Ratio"}]
```

Out[38]=



In[39]:=
```
massCombinedCable[.5, 1 500 000]
```

Out[39]=
```
massCombinedCable[0.5, 1 500 000]
```

For comparison, make a similar function for the Moravec function (rotational acceleration alone, no gravity gradient)

In[43]:=
```
massMoravec[α_, Vc_] := Module[{Vleo},
  Vleo = Sqrt[μ / Rleo];
```
$$\frac{e^{\frac{Vleo^2\,\alpha^2}{2\,Vc^2}}\sqrt{\frac{\pi}{2}}\;Vleo\,\alpha\,\mathrm{Erf}\left[\frac{Vleo\,\alpha}{\sqrt{2}\;Vc}\right]}{Vc}\;\text{/. earthConstants}\Big]$$

In[44]:=
```
massMoravec[.5, V[Spectra]]
```

Out[44]=
```
622.833
```

Finally, a Manipulate widget to get numerical values for the mass ratio as a function of α, material characteristic velocity, and cable length from CG to tip.

In[42]:=
```
Manipulate[
 TableForm[{alpha, MaterialVelocity, CableLength,
   Re[massCombinedCable[alpha, CableLength * 1000, MaterialVelocity]]},
  TableHeadings →
   {{"α", "Matl. Characteristic Vel. m/sec", "Cable Length, km", "Mass Ratio"},
    None}], {{alpha, .75}, .05, 1, .05}, {{MaterialVelocity, 1200}, 1000, 10 000},
 {{CableLength, 1000}, 100, 5000, 100}, SaveDefinitions → True,
 FrameLabel → "Mass Ratio for Spinning Cable"]
```

Out[42]=